Overview of Data Management Tasks

(commands= finan_datamgt.sas)

Make a copy of a SAS data set:

You can use a set statement to make a copy of a data set. In the commands below, the original data set NEWMARCH is created by making an exact copy of MARCH, (assuming the data set MARCH was already created in this session).

data newmarch;

   set march;

run;

Create a subset of data:
You can easily create a subset of your data by using the set statement along with a subsetting if statement. The subsetting if statement acts as a gateway for allowing observations to be written to a data set. It can appear anywhere in the data step. In the examples below, the data set named MARCH15 will contain information on flights on March 15th, 1990, while the data set named LONDON will contain information on all flights to London, and the data set named LONGFLT will contain information on all flights of 1000 miles or more..

data march15;

   set march;

   if date = “15MAR1990”D;

run;

data london;

   set march;

   if dest=”LON”;

run;

data longflt;

   set march;

   if miles >=1000;

run;

Delete cases from a data set:
A delete statement can be used to remove a case or cases from a data set. When the case is deleted, it is permanently removed from the data set. The delete statement takes effect immediately when it is specified, so deleted cases will not be available for any later programming statements.

data shortflt;

   set march;

   if miles >=1000 then delete;

run;

Keep or Drop Variables:
You can control the variables that are included in a SAS data set by using keep and drop statements as part of the data step. The keep and drop statements may be given at any point in the data step, and only take effect at the time the data set is written. 

data march_passngrs;

   set march;

   keep date time orig dest miles boarded transfer nonrev deplane capacity;

run;

data march_passngrs;

   set march;

   drop mail freight;

run;

Create new variables using transformations and recodes:

New variables can be created using transformations and recodes of variables in the data step using assignment statements with SAS functions, or if…then statements. Assignment statements (shown below) are used to create new variables based on the value of previously defined variables, expressions, or constants, and are of the form: 

newvar = expression;

The example below shows how to create new variables using a data step:

data march_recode;

   set march;

   totpassngrs = boarded + transfer + nonrev;

   empty_seats = capacity – totpassngrs;

   totnonpass  = mail + freight;

   pctfull = (totpassngrs/capacity)*100;

   logpassngrs = log(totpassngrs);

   totpassngrs2 = sum(boarded,transfer,nonrev);

   int_pctfull = int(totpassngrs/capacity)*100;

   rnd_pctfull = round(pctfull,.1); 

   if pctfull = 100 then full_flight = 1;

   else full_flight = 0;

   if pctfull = . then full_flight = .;

   if pctfull not=. then do;

      if pctfull < 25 then full_cat = 1;

      if pctfull >=25 and pctfull <50 then full_cat=2;

      if pctfull >=50 and pctfull <75 then full_cat=3;

      if pctfull >=75 then full_cat=4;

   end;

run;  

Check the new variables using Proc Means and Proc Freq:

title "Check New Variables";

proc means data=march_recode;

run;

proc freq data=march_recode;

  tables full_flight full_cat;

run;

                                      Check New Variables     

                                      The MEANS Procedure

 Variable       Label             N           Mean        Std Dev        Minimum        Maximum

 ----------------------------------------------------------------------------------------------

 flight         Flight number   635    447.9086614    275.3102085    114.0000000    982.0000000

 date                           634       11031.98      8.9801263       11017.00       11047.00

 time                           635       47952.47       14707.08       24960.00       75960.00

 miles                          635        1615.25        1338.47    229.0000000        3857.00

 mail                           634    381.0031546     74.6288128    195.0000000    622.0000000

 freight                        634    333.9511041     98.1122248     21.0000000    631.0000000

 boarded                        633    132.3570300     43.4883098     13.0000000    241.0000000

 transfer                       635     14.4062992      5.3362008              0     29.0000000

 nonrev                         635      4.1133858      1.9243731              0      9.0000000

 deplane                        635    146.7842520     45.4289656     18.0000000    250.0000000

 capacity                       635    205.3795276     27.1585929    178.0000000    250.0000000

 totpassngrs                    633    150.8878357     43.0930520     31.0000000    250.0000000

 empty_seats                    633     54.5244866     34.9192529              0    151.0000000

 totnonpass                     633    715.1927330    124.8981261    341.0000000        1085.00

 pctfull                        633     73.0774908     17.7696598     17.2222222    100.0000000

 logpassngrs                    633      4.9681880      0.3292127      3.4339872      5.5214609

 totpassngrs2                   635    150.4598425     43.6959260      9.0000000    250.0000000

 int_pctfull                    633      8.3728278     27.7198922              0    100.0000000

 rnd_pctfull                    633     73.0764613     17.7693610     17.2000000    100.0000000

 full_flight                    633      0.0837283      0.2771989              0      1.0000000

 full_cat                       633      3.3791469      0.6785651      1.0000000      4.0000000

 ----------------------------------------------------------------------------------------------
                                      Check New Variables      

                                       The FREQ Procedure

                                                        Cumulative    Cumulative

                full_flight    Frequency     Percent     Frequency      Percent

                ----------------------------------------------------------------

                          0         580       91.63           580        91.63

                          1          53        8.37           633       100.00

                                     Frequency Missing = 2

                                                      Cumulative    Cumulative

                 full_cat    Frequency     Percent     Frequency      Percent

                 -------------------------------------------------------------

                        1           4        0.63             4         0.63

                        2          59        9.32            63         9.95

                        3         263       41.55           326        51.50

                        4         307       48.50           633       100.00

                                     Frequency Missing = 2
The two variables, TOTPASSNGRS and TOTPASSNGRS2 have different numbers of cases, because they were created in different ways. TOTPASSNGRS is created using the mathematical operators (+), so the resulting variable is missing if any of the variables in the expression is missing. TOTPASSNGRS2 returns the sum of the non-missing argument variables, so if any of the argument variables has a value, the result will have a value.

Also note the syntax used to create the new variables, FULL_FLIGHT and FULL_CAT. When using “Else” with SAS, all other values, including missing will be included in the Else category. We get around this by setting cases with a missing value for PCTFULL to missing in the resulting variable. We use an If…then statement when creating FULL_CAT to be sure this new variable is only created if PCTFULL is not missing (if pctfull not=. then do;). If you use an if…statement, it must be followed by an end statement. 

Sort Cases:
A data set is sorted using Proc Sort. Once sorted, a data set remains sorted, and any later analyses can be done either for the entire data set, or for subgroups by including a by statement in a given procedure.

proc sort data=march_recode;

  by dest;

run;

title “Descriptive Statistics by Destination”;

proc means data=march_recode;

  by dest;

run;

A separate analysis will be done for each of the "by" groups.  

Sorting by more than one variable:

You can sort by several variables, as shown in the example below. Proc sort organizes the data so that the first variable represents the slowest changing index (i.e., cases will be sorted first by DATE, and then by levels of  DEST within DATE). 

proc sort data=march_recode;

     by date dest;

run;

title “Descriptive Statistics by Date and Destination”;

proc means data=march_recode;

     by date dest;

run;

Using the Tagsort Option:

Sorting is one of the more computationally intensive operations that can be done on a SAS data set.  It requires a lot of hard drive space, which can be a problem, especially if the data set has many observations and a large number of variables. You can be more efficient in sorting if you use the tagsort option.  This method basically sorts only the key variables and then rebuilds the dataset by pulling up the appropriate observation and attaching all the rest of the variables. This method of sorting will usually take longer, but uses less hard drive space.  

proc sort data=march_recode tagsort;

     by date dest;

run;

Creating a New Sorted Data Set:

If you wish to create a new data set, and maintain the input data set in its original order, you can use the out= option on the Proc Sort statement, as shown below:

proc sort data=march_recode out=sortdat;

     by date dest;

run;

Selecting cases for analysis:

Cases can be selected for a given analysis by using a where statement. You can select cases based on the value of a character or numeric variable.

Selecting cases based on values of a character variable:

title "Flights to Los Angeles";

proc print data=march_recode;

    where dest = "LAX";

    var flight dest totpassngrs;

run;

If you wish to select the observations to be included in an analysis based on a missing value for a character variable, use quotes around a blank " ", because blank is missing for a character variable.

title "Missing Destination";

proc print data=march_recode;

    where dest = " ";

    var flight dest totpassngrs;

run;

Selecting cases based on values of a numeric variable:

Cases used in an analysis may be selected based on the values of a numeric variable. The Boolean operators (<, >, <=, >=, =, ~=) may be used to get the desired case selection, as shown below:

title "Flights Less than 30 Percent Full";

proc print data=march_recode;

    where pctfull < 30;

    var flight dest totpassngrs capacity pctfull;

run;

Those with PCTFULL missing are also included in this case selection, because missing is evaluated as less than any numeric value.   

The where statement can also be used with “between” to restrict cases used in an analysis. The example below will print those cases with percent full from 25 to 30:

title "Flights Between 25 and 30 Percent Full";

proc print data=march_recode;

    where pctfull between 25 and 30;run;

If you wish to select observations based on a missing value for a numeric variable, use a period to indicate missing, as shown in the example below.  

title "Cases Where Number of Passengers is Missing";

proc print data=march_recode;

    where totpassngrs = .;

    var flight dest passngrs;

run;

Selecting cases based on dates:

You can select cases for a procedure based on dates, by using a SAS date constant.  Note that the date constant is specified in quotes with the day as a two-digit number, followed by a three-letter abbreviation for the month, followed by a 2 or 4-digit number for the year. A letter D (either upper or lower case) must appear after the quote to let SAS know that this is a date.

title "Flights on March 7th, 1990"; 

proc print data=march_recode;

    where date = "07MAR90"D;

run;

You can also use "where … between" with dates to specify a range of dates:

title "Flights March 7th to March 9th , 1990"; 

proc print data=march_recode;

    where date between "07MAR90"D and "09MAR90"D;

run;

You can use the same method for selecting observations based on missing values for a date variable as for a numeric variable, because dates are stored simply as numeric values in SAS.

title "Cases with Missing Date";

proc print data=march_recode;

    where date = .;


    var flight dest date;

run;

